Writing a Simple REST Web Service in PureScript - Part 2

Posted on October 1, 2017 by Abhinav Sarkar
 Tags: purescript, REST, programming, nilenso

② 21 minute read

To recap, in the first part of this two-part tutorial, we built a simple JSON REST web service in PureScript to create, update, get, list and delete users, backed by a Postgres database. In this part we'll work on the rest of the requirements:

- 1. validation of API requests.
- 2. reading the server and database configs from environment variables.
- 3. logging HTTP requests and debugging info.

Contents

- 1. Bugs!
- 2. Validation
- 3. Configuration
- 4. Logging
- 5. Conclusion

1 \$ pulp --watch run

But first.

Bugs!

What happens if we hit a URL on our server which does not exist? Let's fire up the server and test it:

```
1 $ http GET http://localhost:4000/v1/random
2 HTTP/1.1 404 Not Found
3 Connection: keep-alive
  Content-Length: 148
5 Content-Security-Policy: default-src 'self'
6 Content-Type: text/html; charset=utf-8
7 Date: Sat, 30 Sep 2017 08:23:20 GMT
8 X-Content-Type-Options: nosniff
9 X-Powered-By: Express
11 <!DOCTYPE html>
12 <html lang="en">
13 <head>
14 <meta charset="utf-8">
15 <title>Error</title>
16 </head>
17 <body>
18 Cannot GET /v1/random
```

```
19 </body>
20 </html>
```

We get back a default HTML response with a 404 status from Express. Since we are writing a JSON API, we should return a JSON response in this case too. We add the following code in the src/SimpleService/Server.purs file to add a catch-all route and send a 404 status with a JSON error message:

```
1 -- previous code
2 import Data.Either (fromRight)
3 import Data.String.Regex (Regex, regex) as Re
4 import Data.String.Regex.Flags (noFlags) as Re
  import Node.Express.App (App, all, delete, get, http, listenHttp, post, useExternal)
6 import Node.Express.Response (sendJson, setStatus)
 import Partial.Unsafe (unsafePartial)
8 -- previous code
10 allRoutePattern :: Re.Regex
11 allRoutePattern = unsafePartial $ fromRight $ Re.regex "/.*" Re.noFlags
13 app :: forall eff. PG.Pool -> App (postgreSQL :: PG.POSTGRESQL | eff)
14 app pool = do
15
    useExternal jsonBodyParser
16
17
    delete "/v1/user/:id" $ deleteUser pool
18
    19
20
    patch "/v1/user/:id" $ updateUser pool
    21
22
   all allRoutePattern do
23
24
     setStatus 404
25
    sendJson {error: "Route not found"}
26
  where
      patch = http (CustomMethod "patch")
27
```

allRoutePattern matches all routes because it uses a "/.*" regular expression. We place it as the last route to match all the otherwise unrouted requests. Let's see what is the result:

```
1  $ http GET http://localhost:4000/v1/random
2  HTTP/1.1  404  Not  Found
3  Connection: keep-alive
4  Content-Length: 27
5  Content-Type: application/json; charset=utf-8
6  Date: Sat, 30  Sep  2017  08:46:46  GMT
7  ETag: W/"1b-772e0u4nrE48ogbR0KmKfSvrHUE"
8  X-Powered-By: Express
9
10  {
11    "error": "Route not found"
12 }
```

Now we get a nicely formatted JSON response.

Another scenario is when our application throws some uncaught error. To simulate this, we shut down our postgres database and hit the server for listing users:

```
1  $ http GET http://localhost:4000/v1/users
2 HTTP/1.1 500 Internal Server Error
3 Connection: keep-alive
4 Content-Length: 372
5 Content-Security-Policy: default-src 'self'
6 Content-Type: text/html; charset=utf-8
7 Date: Sat, 30 Sep 2017 08:53:40 GMT
8 X-Content-Type-Options: nosniff
9 X-Powered-By: Express
10
11 <!DOCTYPE html>
12 <html lang="en">
13 <head>
14 <meta charset="utf-8">
15 <title>Error</title>
16 </head>
17 <body>
18 Frror: connect ECONNREFUSED 127.0.0.1:5432<br > &nbsp; &nbsp; at
   Object._errnoException (util.js:1026:11)<br/>
onbsp; &nbsp;at _exceptionWithHostPort
  (util.js:1049:20)<br> &nbsp; &nbsp;at TCPConnectWrap.afterConnect [as oncomplete]
   (net.js:1174:14)
19 </body>
20 </html>
```

We get another default HTML response from Express with a 500 status. Again, in this case we'd like to return a JSON response. We add the following code to the src/SimpleService/Server.purs file:

```
1 -- previous code
2 import Control.Monad.Eff.Exception (message)
import Node.Express.App (App, all, delete, get, http, listenHttp, post, useExternal,
  useOnError)
4 -- previous code
6 app :: forall eff. PG.Pool -> App (postgreSQL :: PG.POSTGRESQL | eff)
7 app pool = do
   -- previous code
8
   useOnError \err -> do
9
10
     setStatus 500
      sendJson {error: message err}
11
12 where
13
      patch = http (CustomMethod "patch")
```

We add the useOnError handler which comes with purescript-express to return the error message as a JSON response. Back on the command-line:

```
1  $ http GET http://localhost:4000/v1/users
2  HTTP/1.1 500 Internal Server Error
3  Connection: keep-alive
4  Content-Length: 47
5  Content-Type: application/json; charset=utf-8
6  Date: Sat, 30 Sep 2017 09:01:37 GMT
7  ETag: W/"2f-cJuIW6961YCpo9TWDSZ9VWHLGHE"
8  X-Powered-By: Express
9
10 {
```

```
"error": "connect ECONNREFUSED 127.0.0.1:5432"
12 }
```

It works! Bugs are fixed now. We proceed to add next features.

Validation

Let's recall the code to update a user from the src/SimpleService/Handler.purs file:

```
updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
   updateUser pool = getRouteParam "id" >>= case _ of
     Nothing -> respond 422 { error: "User ID is required" }
4
     Just sUserId -> case fromString sUserId of
       Nothing -> respond 422 { error: "User ID must be positive: " <> sUserId }
5
6
       Just userId -> getBody >>= case _ of
         Left errs -> respond 422 { error: intercalate ", " $ map renderForeignError
   errs}
8
         Right (UserPatch userPatch) -> case unNullOrUndefined userPatch.name of
           Nothing -> respondNoContent 204
9
           Just userName -> if userName == ""
10
             then respond 422 { error: "User name must not be empty" }
11
12
               savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction
13
   conn do
14
                 P.findUser conn userId >>= case _ of
                   Nothing -> pure Nothing
15
                   Just (User user) -> do
16
17
                     let user' = User (user { name = userName })
18
                     P.updateUser conn user'
                     pure $ Just user'
19
20
               case savedUser of
                 Nothing -> respond 404 { error: "User not found with id: " <> sUserId }
21
22
                 Just user -> respond 200 (encode user)
```

As we can see, the actual request handling logic is obfuscated by the request validation logic for the user id and the user name patch parameters. We also notice that we are using three constructs for validation here: Maybe, Either and if-then-else. However, we can use just Either to subsume all these cases as it can "carry" a failure as well as a success case. Either also comes with a nice monad transformer ExceptT which provides the do syntax for failure propagation. So we choose ExceptT as the base construct for our validation framework and write functions to upgrade Maybe and if-then-else to it. We add the following code to the src/SimpleService/Validation.purs file:

```
module SimpleService.Validation

(module MoreExports, module SimpleService.Validation) where

import Prelude

import Control.Monad.Except (ExceptT, except, runExceptT)

import Data.Either (Either(..))

import Data.Maybe (Maybe(..))

import Node.Express.Handler (HandlerM, Handler)

import Node.Express.Response (sendJson, setStatus)

import Node.Express.Types (EXPRESS)

import Control.Monad.Except (except) as MoreExports

import Node.Express.Types
```

```
14 type Validation eff a = ExceptT String (HandlerM (express :: EXPRESS | eff)) a
15
16 exceptMaybe :: forall e m a. Applicative m => e -> Maybe a -> ExceptT e m a
17 exceptMaybe e a = except $ case a of
18
     Just x -> Right x
19
     Nothing -> Left e
20
21 exceptCond :: forall e m a. Applicative m => e -> (a -> Boolean) -> a -> ExceptT e m a
22 exceptCond e cond a = except $ if cond a then Right a else Left e
23
24 withValidation :: forall eff a. Validation eff a -> (a -> Handler eff) -> Handler eff
25 withValidation action handler = runExceptT action >>= case _ of
    Left err -> do
       setStatus 422
27
28
       sendJson {error: err}
     Right x -> handler x
29
```

We re-export except from the Control.Monad.Except module. We also add a withValidation function which runs an ExceptT based validation and either returns an error response with a 422 status in case of a failed validation or runs the given action with the valid value in case of a successful validation.

Using these functions, we now write updateUser in the src/SimpleService/Handler.purs file as:

```
1 -- previous code
2 import Control.Monad.Trans.Class (lift)
3 import Data.Bifunctor (lmap)
  import Data.Foreign (ForeignError, renderForeignError)
5 import Data.List.NonEmpty (toList)
6 import Data.List.Types (NonEmptyList)
  import Data.Tuple (Tuple(...))
7
8 import SimpleService.Validation as V
9
   -- previous code
10
11 renderForeignErrors :: forall a. Either (NonEmptyList ForeignError) a -> Either String
12 renderForeignErrors = lmap (toList >>> map renderForeignError >>> intercalate ", ")
13
14 updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
updateUser pool = V.withValidation (Tuple <$> getUserId <*> getUserPatch)
16
                                      \(Tuple userId (UserPatch userPatch)) ->
       case unNullOrUndefined userPatch.name of
17
18
         Nothing -> respondNoContent 204
19
         Just uName -> V.withValidation (getUserName uName) \userName -> do
20
           savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn
   do
21
             P.findUser conn userId >>= case of
               Nothing -> pure Nothing
22
23
               Just (User user) -> do
24
                 let user' = User (user { name = userName })
25
                 P.updateUser conn user'
                 pure $ Just user'
26
27
           case savedUser of
             Nothing -> respond 404 { error: "User not found with id: " <> show userId }
28
29
             Just user -> respond 200 (encode user)
30
       getUserId = lift (getRouteParam "id")
31
```

The validation logic has been extracted out in separate functions now which are composed using Applicative. The validation steps are composed using the ExceptT monad. We are now free to express the core logic of the function clearly. We rewrite the src/SimpleService/Handler.purs file using the validations:

```
module SimpleService.Handler where
3
  import Prelude
4
  import Control.Monad.Aff.Class (liftAff)
6 import Control.Monad.Trans.Class (lift)
7
  import Data.Bifunctor (lmap)
8 import Data.Either (Either)
9 import Data.Foldable (intercalate)
10 import Data.Foreign (ForeignError, renderForeignError)
import Data.Foreign.Class (encode)
import Data.Foreign.NullOrUndefined (unNullOrUndefined)
13 import Data.Int (fromString)
14 import Data.List.NonEmpty (toList)
15 import Data.List.Types (NonEmptyList)
16 import Data.Maybe (Maybe(...))
17 import Data.Tuple (Tuple(..))
18 import Database.PostgreSQL as PG
19 import Node.Express.Handler (Handler)
20 import Node.Express.Request (getBody, getRouteParam)
21 import Node.Express.Response (end, sendJson, setStatus)
22 import SimpleService.Persistence as P
23 import SimpleService. Validation as V
24 import SimpleService.Types
25
26 getUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
27 getUser pool = V.withValidation getUserId \userId ->
     liftAff (PG.withConnection pool $ flip P.findUser userId) >>= case of
       Nothing -> respond 404 { error: "User not found with id: " <> show userId }
29
       Just user -> respond 200 (encode user)
30
31
32 deleteUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
33 deleteUser pool = V.withValidation getUserId \userId -> do
34
     found <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn do
35
       P.findUser conn userId >>= case of
36
         Nothing -> pure false
37
         Just -> do
           P.deleteUser conn userId
38
39
           pure true
     if found
40
41
       then respondNoContent 204
       else respond 404 { error: "User not found with id: " <> show userId }
42
43
44 createUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
```

```
45 createUser pool = V.withValidation getUser \user@(User ) -> do
     liftAff (PG.withConnection pool $ flip P.insertUser user)
46
47
     respondNoContent 201
48
     where
49
       getUser = lift getBody
50
         >>= V.except <<< renderForeignErrors
         >>= V.exceptCond "User ID must be positive" (\(\(\begin{align*} \text{User user}\)) -> user.id > 0\)
51
         >>= V.exceptCond "User name must not be empty" (\(\(\text{User user}\)) -> user.name /= "")
52
53
   updateUser :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
54
   updateUser pool = V.withValidation (Tuple <$> getUserId <*> getUserPatch)
56
                                        \(Tuple userId (UserPatch userPatch)) ->
57
       case unNullOrUndefined userPatch.name of
58
         Nothing -> respondNoContent 204
59
         Just uName -> V.withValidation (getUserName uName) \userName -> do
           savedUser <- liftAff $ PG.withConnection pool \conn -> PG.withTransaction conn
60
   do
             P.findUser conn userId >>= case _ of
61
               Nothing -> pure Nothing
62
63
               Just (User user) -> do
                 let user' = User (user { name = userName })
64
                 P.updateUser conn user'
65
66
                 pure $ Just user'
           case savedUser of
68
             Nothing -> respond 404 { error: "User not found with id: " <> show userId }
              Just user -> respond 200 (encode user)
69
70
     where
71
       getUserPatch = lift getBody >>= V.except <<< renderForeignErrors</pre>
72
       getUserName = V.exceptCond "User name must not be empty" (_ /= "")
73
74 listUsers :: forall eff. PG.Pool -> Handler (postgreSQL :: PG.POSTGRESQL | eff)
75 listUsers pool = liftAff (PG.withConnection pool P.listUsers) >>= encode >>> respond
   200
76
77 getUserId :: forall eff. V. Validation eff Int
78 getUserId = lift (getRouteParam "id")
    >>= V.exceptMaybe "User ID is required"
79
     >>= fromString >>> V.exceptMaybe "User ID must be an integer"
     >>= V.exceptCond "User ID must be positive" (_ > 0)
81
82
83 renderForeignErrors :: forall a. Either (NonEmptyList ForeignError) a -> Either String
84 renderForeignErrors = lmap (toList >>> map renderForeignError >>> intercalate ", ")
86 respond :: forall eff a. Int -> a -> Handler eff
87 respond status body = do
   setStatus status
88
89
     sendJson body
90
91 respondNoContent :: forall eff. Int -> Handler eff
92 respondNoContent status = do
93
     setStatus status
94
     end
```

The code is much cleaner now. Let's try out a few test cases:

```
2 HTTP/1.1 201 Created
3 Connection: keep-alive
4 Content-Length: 0
5 Date: Sat, 30 Sep 2017 12:13:37 GMT
6 X-Powered-By: Express
1  $ http POST http://localhost:4000/v1/users id:=3
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 102
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:13:50 GMT
7 ETag: W/"66-/c4cfoquQZGwtDBUzHjJydJAHJ0"
8 X-Powered-By: Express
10 {
       "error": "Error at array index 0: (ErrorAtProperty \"name\" (TypeMismatch
   \"String\" \"Undefined\"))"
12 }
1 $ http POST http://localhost:4000/v1/users id:=3 name=""
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 39
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:02 GMT
7 ETag: W/"27-JQsh12xu/rEFdWy8REF4NMtBUB4"
8 X-Powered-By: Express
9
10 {
       "error": "User name must not be empty"
11
12 }
1 $ http POST http://localhost:4000/v1/users id:=0 name=roger
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 36
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:14 GMT
7 ETag: W/"24-Pvt1L4eGilBmVtaOGHlSReJ413E"
8 X-Powered-By: Express
9
10 {
11
      "error": "User ID must be positive"
12 }
1  $ http GET http://localhost:4000/v1/user/3
2 HTTP/1.1 200 OK
3 Connection: keep-alive
4 Content-Length: 23
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:28 GMT
7 ETag: W/"17-1scpiB1FT9DBu9s4I1gNWSjH2go"
8 X-Powered-By: Express
```

```
9
10 {
11
       "id": 3,
12
       "name": "roger"
13 }
1  $ http GET http://localhost:4000/v1/user/asdf
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 38
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:40 GMT
7 ETag: W/"26-//tvORl1gGDUMwgSaqbEpJhuadI"
8 X-Powered-By: Express
9
10 {
       "error": "User ID must be an integer"
11
12 }
1  $ http GET http://localhost:4000/v1/user/-1
2 HTTP/1.1 422 Unprocessable Entity
3 Connection: keep-alive
4 Content-Length: 36
5 Content-Type: application/json; charset=utf-8
6 Date: Sat, 30 Sep 2017 12:14:45 GMT
7 ETag: W/"24-Pvt1L4eGilBmVtaOGHlSReJ413E"
8 X-Powered-By: Express
10 {
       "error": "User ID must be positive"
11
12 }
```

It works as expected.

Configuration

Right now our application configuration resides in the main function:

```
main = runServer port databaseConfig
2
     where
3
       port = 4000
       databaseConfig = { user: "abhinav"
4
5
                          , password: ""
                          , host: "localhost"
7
                          , port: 5432
8
                          , database: "simple_service"
9
                          , max: 10
10
                          , idleTimeoutMillis: 1000
11
                         }
```

We are going to extract it out of the code and read it from the environment variables using the purescript-config package. First, we install the required packages using bower.

```
1 $ bower install --save purescript-node-process purescript-config
```

Now, we write the following code in the src/SimpleService/Config.purs file:

```
module SimpleService.Config where
3 import Data.Config
4 import Prelude
6 import Control.Monad.Eff (Eff)
7 import Data.Config.Node (fromEnv)
8 import Data.Either (Either)
9 import Data.Set (Set)
10 import Database.PostgreSQL as PG
import Node.Process (PROCESS)
13 type ServerConfig =
14
   { port
             :: Int
     , databaseConfig :: PG.PoolConfiguration
15
16
    }
17
18 databaseConfig :: Config {name :: String} PG.PoolConfiguration
19 databaseConfig =
20 { user: _, password: _, host: _, port: _, database: _, max: _, idleTimeoutMillis: _
  }
21
   <$> string {name: "user"}
  <*> string {name: "password"}
   <*> string {name: "host"}
23
  <*> int {name: "port"}
24
25 <*> string {name: "database"}
  <*> int {name: "pool size"}
26
    27
28
29 portConfig :: Config {name :: String} Int
30 portConfig = int {name: "port"}
31
32 serverConfig :: Config {name :: String} ServerConfig
33 serverConfig =
  { port: _, databaseConfig: _}
34
   <$> portConfig
35
    <*> prefix {name: "db"} databaseConfig
36
37
38 readServerConfig :: forall eff.
                     Eff (process :: PROCESS | eff) (Either (Set String) ServerConfig)
40 readServerConfig = fromEnv "SS" serverConfig
```

We use the applicative DSL provided in Data.Config module to build a description of our configuration. This description contains the keys and types of the configuration, for consumption by various interpreters. Then we use the fromEnv interpreter to read the config from the environment variables derived from the name fields in the records in the description in the readServerConfig function. We also write a bash script to set those environment variables in the development environment in the setenv.sh file:

```
1 export SS_PORT=4000
2 export SS_DB_USER="abhinav"
3 export SS_DB_PASSWORD=""
4 export SS_DB_HOST="localhost"
5 export SS_DB_PORT=5432
```

```
6 export SS_DB_DATABASE="simple_service"
7 export SS_DB_POOL_SIZE=10
8 export SS_DB_IDLE_CONN_TIMEOUT_MILLIS=1000
```

Now we rewrite our src/Main.purs file to use the readServerConfig function:

```
module Main where
3
  import Prelude
4
5 import Control.Monad.Eff (Eff)
6 import Control.Monad.Eff.Console (CONSOLE, log)
7 import Data.Either (Either(..))
8 import Data.Set (toUnfoldable)
9 import Data.String (joinWith)
10 import Database.PostgreSQL as PG
import Node.Express.Types (EXPRESS)
12 import Node.Process (PROCESS)
13 import Node.Process as Process
14 import SimpleService.Config (readServerConfig)
15 import SimpleService.Server (runServer)
16
17 main :: forall eff. Eff ( console :: CONSOLE
18
                           , express :: EXPRESS
19
                           , postgreSQL :: PG.POSTGRESQL
                           , process :: PROCESS
20
                           | eff ) Unit
22 main = readServerConfig >>= case of
23
   Left missingKeys -> do
      log $ "Unable to start. Missing Env keys: " <> joinWith ", " (toUnfoldable
   missingKeys)
25
       Process.exit 1
     Right { port, databaseConfig } -> runServer port databaseConfig
```

If readServerConfig fails, we print the missing keys to the console and exit the process. Else we run the server with the read config.

To test this, we stop the server we ran in the beginning, source the config, and run it again:

```
$ pulp --watch run

* Building project in /Users/abhinav/ps-simple-rest-service

* Build successful.

Server listening on :4000

C

$ source setenv.sh

pulp --watch run

Building project in /Users/abhinav/ps-simple-rest-service

Build successful.

Server listening on :4000
```

It works! We test the failure case by opening another terminal which does not have the environment variables set:

```
1 $ pulp run
2 * Building project in /Users/abhinav/ps-simple-rest-service
```

```
3 * Build successful.
4 Unable to start. Missing Env keys: SS_DB_DATABASE, SS_DB_HOST,
    SS_DB_IDLE_CONN_TIMEOUT_MILLIS, SS_DB_PASSWORD, SS_DB_POOL_SIZE, SS_DB_PORT,
    SS_DB_USER, SS_PORT
5 * ERROR: Subcommand terminated with exit code 1
```

Up next, we add logging to our application.

Logging

For logging, we use the purescript-logging package. We write a logger which logs to stdout; in the src/SimpleService/Logger.purs file:

```
module SimpleService.Logger
2
    ( debug
3
     , info
    , warn
     , error
6
     ) where
7
8 import Prelude
10 import Control.Logger as L
import Control.Monad.Eff.Class (class MonadEff, liftEff)
12 import Control.Monad.Eff.Console as C
import Control.Monad.Eff.Now (NOW, now)
14 import Data.DateTime.Instant (toDateTime)
15 import Data.Either (fromRight)
import Data.Formatter.DateTime (Formatter, format, parseFormatString)
17 import Data.Generic.Rep (class Generic)
18 import Data.Generic.Rep.Show (genericShow)
19 import Data.String (toUpper)
20 import Partial.Unsafe (unsafePartial)
21
22 data Level = Debug | Info | Warn | Error
24 derive instance eqLevel :: Eq Level
25 derive instance ordLevel :: Ord Level
26 derive instance genericLevel :: Generic Level _
27
28 instance showLevel :: Show Level where
     show = toUpper <<< genericShow</pre>
30
31 type Entry =
   { level :: Level
32
33
     , message :: String
34
     }
36 dtFormatter :: Formatter
37 dtFormatter = unsafePartial $ fromRight $ parseFormatString "YYYY-MM-DD HH:mm:ss.SSS"
38
39 logger :: forall m e. (
            MonadEff (console :: C.CONSOLE, now :: NOW | e) m) => L.Logger m Entry
41 logger = L.Logger $ \{ level, message } -> liftEff do
   time <- toDateTime <$> now
42
     C.log $ "[" <> format dtFormatter time <> "] " <> show level <> " " <> message
43
```

```
44
45 log :: forall m e.
46
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m
        => Entry -> m Unit
47
48 log entry@{level} = L.log (L.cfilter (\e -> e.level == level) logger) entry
50 debug :: forall m e.
            MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
51
52 debug message = log { level: Debug, message }
53
54 info :: forall m e.
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
56 info message = log { level: Info, message }
57
58 warn :: forall m e.
           MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
59
60 warn message = log { level: Warn, message }
62 error :: forall m e.
63
            MonadEff (console :: C.CONSOLE , now :: NOW | e) m => String -> m Unit
64 error message = log { level: Error, message }
```

purescript-logging lets us define our own logging levels and loggers. We define four log levels, and a log entry type with the log level and the message. Then we write the logger which will print the log entry to stdout along with the current time as a well formatted string. We define convenience functions for each log level.

Before we proceed, let's install the required dependencies.

```
1 $ bower install --save purescript-logging purescript-now purescript-formatters
```

Now we add a request logger middleware to our server in the src/SimpleService/Server.purs file:

```
1 -- previous code
2 import Control.Monad.Eff.Console (CONSOLE)
3 import Control.Monad.Eff.Now (NOW)
4 import Data.Maybe (maybe)
5 import Data.String (toUpper)
6 import Node. Express. App (App, all, delete, get, http, listenHttp, post, use,
   useExternal, useOnError)
7 import Node.Express.Handler (Handler, next)
8 import Node.Express.Request (getMethod, getPath)
9 import SimpleService.Logger as Log
10 -- previous code
12 requestLogger :: forall eff. Handler (console :: CONSOLE, now :: NOW | eff)
13 requestLogger = do
14
    method <- getMethod
          <- getPath
15
     path
     Log.debug $ "HTTP: " <> maybe "" id ((toUpper <<< show) <$> method) <> " " <> path
16
17
     next
18
19 app :: forall eff.
         PG.Pool
20
21
       -> App (postgreSQL :: PG.POSTGRESQL, console :: CONSOLE, now :: NOW | eff)
22 app pool = do
```

```
useExternal jsonBodyParser
use requestLogger
-- previous code
```

We also convert all our previous logging statements which used Console.log to use SimpleService.Logger and add logs in our handlers. We can see logging in effect by restarting the server and hitting it:

```
1  $ pulp --watch run
2  * Building project in /Users/abhinav/ps-simple-rest-service
3  * Build successful.
4  [2017-09-30 16:02:41.634] INFO Server listening on :4000
5  [2017-09-30 16:02:43.494] DEBUG HTTP: PATCH /v1/user/3
6  [2017-09-30 16:02:43.517] DEBUG Updated user: 3
7  [2017-09-30 16:03:46.615] DEBUG HTTP: DELETE /v1/user/3
8  [2017-09-30 16:03:46.635] DEBUG Deleted user 3
9  [2017-09-30 16:05:03.805] DEBUG HTTP: GET /v1/users
```

Conclusion

In this tutorial we learned how to create a simple JSON REST web service written in PureScript with persistence, validation, configuration and logging. The complete code for this tutorial can be found in github. This post can be discussed on r/purescript.

© 2017, Abhinav Sarkar

Generated by Hakyll | Source